

Functional Programming

Kung Chen and Shin-Cheng Mu

FLOLAC, 2016

A Quick Introduction to Haskell

- We will mostly learn some syntactical issues, but there are some important messages too.
- Most of the materials today are adapted from the book *Introduction to Functional Programming using Haskell* by Richard Bird. Prentice Hall 1998.
- References to more Haskell materials are on the course homepage.

Course Materials and Tools

- Course homepage: <http://flolac.iis.sinica.edu.tw/pl2015>
 - Announcements, slides, assignments, additional materials, etc.
- We will be using the Glasgow Haskell Compiler (GHC).
 - A Haskell compiler written in Haskell, with an interpreter that both interprets and runs compiled code.

Function Definition

- A function definition consists of a type declaration, and the definition of its body:

```
square    :: Int → Int
square x  = x × x
```

```
smaller   :: Int → Int → Int
smaller x y = if x ≤ y then x else y
```

- The GHCi interpreter evaluates expressions in the loaded context:

```
? square 3768
14197824
? square (smaller 5 (3 + 4))
25
```

1 Values and Evaluation

Evaluation

One possible sequence of evaluating (simplifying, or reducing) *square* (3 + 4):

```
square (3 + 4)
= { definition of + }
square 7
= { definition of square }
7 × 7
= { definition of × }
49
```

Another Evaluation Sequence

- Another possible reduction sequence:

```
square (3 + 4)
= { definition of square }
(3 + 4) × (3 + 4)
= { definition of + }
7 × (3 + 4)
= { definition of + }
7 × 7
= { definition of × }
49
```

- In this sequence the rule for *square* is applied first. The final result stays the same.
- Do different evaluations orders always yield the same thing?

A Non-terminating Reduction

- Consider the following program:

```
three  :: Int → Int
three x = 3
infinity :: Int
infinity = infinity + 1
```

- Try evaluating `three infinity`. If we simplify `infinity` first:

```
three infinity
= { definition of infinity }
three (infinity + 1)
= three ((infinity + 1) + 1)...
```

- If we start with simplifying `three`:

```
three infinity
= { definition of three }
3
```

Evaluation Order

- There can be many other evaluation orders. As we have seen, some terminates while some do not.
- *normal form*: an expression that cannot be reduced anymore.
 - 49 is in normal form, while 7×7 is not.
 - Some expressions do not have a normal form. E.g. `infinity`.
- A corollary of the *Church–Rosser theorem*: an expression has at most one normal form.
 - If two evaluation sequences both terminate, they reach the same normal form.

Evaluation Order

- Applicative order evaluation: starting with the innermost reducible expression (a redex).
- Normal order evaluation: starting with the outermost redex.
- If an expression has a normal form, normal order evaluation delivers it. Hence the name.
- For now you can imagine that Haskell uses normal order evaluation. A way to implement normal order evaluation is called *lazy evaluation*.

2 Functions

Mathematical Functions

- Mathematically, a function is a mapping between arguments and results.
 - A function $f :: A \rightarrow B$ maps each element in A to a unique element in B .
- In contrast, C “functions” are not mathematical functions:
 - `int y = 1; int f (x:int) { return ((y++) * x); }`
- Functions in Haskell have no such *side-effects*: (unconstrained) assignments, IO, etc.
- Why removing these useful features? We will talk about that later in this course.

2.1 Using Functions

Curried Functions

- Consider again the function `smaller`:

```
smaller  :: Int → Int → Int
smaller x y = if x ≤ y then x else y
```

- We sometimes informally call it a function “taking two arguments”.
- Usage: `smaller 3 4`.
- Strictly speaking, however, `smaller` is a function returning a function. The type should be bracketed as $Int \rightarrow (Int \rightarrow Int)$.

Precedence and Association

- In a sense, all Haskell functions takes exactly one argument.
 - Such functions are often called *curried*.
- Type: $a \rightarrow b \rightarrow c = a \rightarrow (b \rightarrow c)$, not $(a \rightarrow b) \rightarrow c$.
- Application: $f x y = (f x) y$, not $f (x y)$.
 - `smaller 3 4` means `(smaller 3) 4`.
 - `square square 3` means `(square square) 3`, which results in a type error.
- Function application binds tighter than infix operators. E.g. `square 3 + 4` means `(square 3) + 4`.

Why Currying?

- It exposes more chances to reuse a function, since it can be partially applied.

```
twice    :: (a -> a) -> (a -> a)
twice f x = f (f x)
quad     :: Int -> Int
quad     = twice square
```

- Try evaluating `quad 3`:

```
quad 3
= twice square 3
= square (square 3)
= ...
```

- Had we defined:

```
twice    :: (a -> a, a) -> a
twice (f, x) = f (f x)
```

we would have to write

```
quad    :: Int -> Int
quad x = twice (square, x)
```

- There are situations where you'd prefer not to have curried functions. We will talk about conversion between curried and uncurried functions later.

2.2 Sectioning

Sectioning

- Infix operators are curried too. The operator `(+)` may have type `Int -> Int -> Int`.
- Infix operator can be partially applied too.

```
(x ⊕) y = x ⊕ y
(⊕ y) x = x ⊕ y
```

- `(1 +)` :: `Int -> Int` increments its argument by one.
- `(1.0 /)` :: `Float -> Float` is the “reciprocal” function.
- `(/ 2.0)` :: `Float -> Float` is the “halving” function.

Infix and Prefix

- To use an infix operator in prefix position, surrounded it in parentheses. For example, `(+) 3 4` is equivalent to `3 + 4`.
- Surround an ordinary function by back-quotes (not quotes!) to put it in infix position. E.g. `3 'mod' 4` is the same as `mod 3 4`.

Function Composition

- Functions composition:

```
(.) :: (b -> c) -> (a -> b) -> (a -> c)
(f . g) x = f (g x)
```

- E.g. another way to write `quad`:

```
quad :: Int -> Int
quad = square . square
```

- Some important properties:

- `id . f = f = f . id`, where `id x = x`.
- `(f . g) . h = f . (g . h)`.

2.3 Definitions

Guarded Equations

- Recall the definition:

```
smaller    :: Int -> Int -> Int
smaller x y = if x ≤ y then x else y
```

- We can also write:

```
smaller    :: Int -> Int -> Int
smaller x y | x ≤ y = x
            | x > y = y
```

- Equivalently,

```
smaller :: Int -> Int -> Int
smaller x y | x ≤ y      = x
            | otherwise = y
```

- Helpful when there are many choices:

```
signum :: Int -> Int
signum x | x > 0 = 1
         | x == 0 = 0
         | x < 0 = -1
```

Otherwise we'd have to write

```
signum x = if x > 0 then 1
          else if x == 0 then 0 else -1
```

λ Expressions

- Since functions are first-class constructs, we can also construct functions in expressions.
- A λ expression denotes an anonymous function.
 - $\lambda x \rightarrow e$: a function with argument x and body e .
 - $\lambda x \rightarrow \lambda y \rightarrow e$ abbreviates to $\lambda x y \rightarrow e$.
 - In ASCII, we write λ as `\`
- Yet another way to define *smaller*:
$$\begin{aligned} \text{smaller} &:: \text{Int} \rightarrow \text{Int} \rightarrow \text{Int} \\ \text{smaller} &= \lambda x y \rightarrow \text{if } x \leq y \text{ then } x \text{ else } y \end{aligned}$$
- Why λ s? Sometimes we may want to quickly define a function and use it only once.
- In fact, λ is a more primitive concept.

Local Definitions

There are two ways to define local bindings in Haskell.

- **let**-expression:

$$\begin{aligned} f &:: \text{Float} \rightarrow \text{Float} \rightarrow \text{Float} \\ f \ x \ y &= \text{let } a = (x + y)/2 \\ &\quad b = (x + y)/3 \\ &\quad \text{in } (a + 1) \times (b + 2) \end{aligned}$$

- **where**-clause:

$$\begin{aligned} f &:: \text{Int} \rightarrow \text{Int} \rightarrow \text{Int} \\ f \ x \ y &| x \leq 10 = x + a \\ &| x > 10 = x - a \\ &\quad \text{where } a = \text{square } (y + 1) \end{aligned}$$

- **let** can be used in expressions (e.g. `1 + (let..in..)`), while **where** qualifies multiple guarded equations.

3 Types

Types

- The universe of values is partitioned into collections, called *types*.
- Some basic types: *Int*, *Float*, *Bool*, *Char*...

- Type “constructors”: functions, lists, trees ... to be introduced later.
- Operations on values of a certain type might not make sense for other types. For example: `square square 3`.
- Strong typing: the type of a well-formed expression can be deduced from the constituents of the expression.
 - It helps you to detect errors.
 - More importantly, programmers may consider the types for the values being defined before considering the definition themselves, leading to clear and well-structured programs.

Polymorphic Types

- Suppose `square :: Int → Int` and `sqrt :: Int → Float`.
 - `square · square :: Int → Int`
 - `sqrt · square :: Int → Float`
- The `(.)` operator has different types in the two expressions:
 - `(.) :: (Int → Int) → (Int → Int) → (Int → Int)`
 - `(.) :: (Int → Float) → (Int → Int) → (Int → Float)`
- To allow `(.)` to be used in many situations, we introduce type variables and let its type be: $(b \rightarrow c) \rightarrow (a \rightarrow b) \rightarrow (a \rightarrow c)$.

Summary So Far

- Functions are essential building blocks in a Haskell program. They can be applied, composed, passed as arguments, and returned as results.
- Types sometimes guide you through the design of a program.
- Equational reasoning: let the symbols do the work!

Recommended Textbooks

- *Introduction to Functional Programming using Haskell* [Bir98]. My recommended book. Covers equational reasoning very well.
- *Programming in Haskell* [Hut07]. A thin but complete textbook.

Online Haskell Tutorials

- *Learn You a Haskell for Great Good!* [Lip11], a nice tutorial with cute drawings!
- *Yet Another Haskell Tutorial* [DI02].
- *A Gentle Introduction to Haskell* by Paul Hudak, John Peterson, and Joseph H. Fasel: a bit old, but still worth a read. [HPF00]
- *Real World Haskell* [OSG98]. Freely available online. It assumes some basic knowledge of Haskell, however.

4 Simple Datatypes

4.1 Booleans

Booleans

The datatype *Bool* can be introduced with a *datatype declaration*:

```
data Bool = False | True
```

(But you need not do so. The type *Bool* is already defined in the Haskell Prelude.)

Datatype Declaration

- In Haskell, a **data** declaration defines a new type.

```
data Type = Con1 Type11 Type12...
          | Con2 Type21 Type22...
          | :
```

- The declaration above introduces a new type, *Type*, with several cases.
- Each case starts with a constructor, and several (zero or more) arguments (also types).
- Informally it means “a value of type *Type* is either a *Con₁* with arguments *Type₁₁*, *Type₁₂*, ..., or a *Con₂* with arguments *Type₂₁*, *Type₂₂*, ...”
- Types and constructors begin in capital letters.

Functions on Booleans

Negation:

```
not      :: Bool → Bool
not False = True
not True  = False
```

- Notice the definition by *pattern matching*. The definition has two cases, because *Bool* is defined by two cases. The shape of the function follows the shape of its argument.

Functions on Booleans

Conjunction and disjunction:

```
(&&), (||) :: Bool → Bool → Bool
False && x = False
True  && x = x
False || x = x
True  || x = True
```

Functions on Booleans

Equality check:

```
(==), (≠) :: Bool → Bool → Bool
x == y    = (x && y) || (not x && not y)
x ≠ y     = not (x == y)
```

- `=` is a definition, while `==` is a function.
- `≠` is written `/=` in ASCII.

Example

```
leapyear :: Int → Bool
leapyear y = (y 'mod' 4 == 0) &&
              (y 'mod' 100 ≠ 0 || y 'mod' 400 == 0)
```

- Note: *y 'mod' 100* could be written *mod y 100*. The backquotes turns an ordinary function to an infix operator.
- It's just personal preference whether to do so.

4.2 Characters

Characters

- You can think of *Char* as a big **data** definition:

```
data Char = 'a' | 'b' | ...
```

with functions:

```
ord :: Char → Int
chr :: Int → Char
```

- Characters are compared by their order:

```
isDigit :: Char → Bool
isDigit x = '0' ≤ x && x ≤ '9'
```

Equality Check

- Of course, you can test equality of characters too:

```
(==) :: Char → Char → Bool
```

- `(==)` is an *overloaded* name — one name shared by many different definitions of equalities, for different types:

```
- (==) :: Int → Int → Bool
- (==) :: (Int, Char) → (Int, Char) → Bool
- (==) :: [Int] → [Int] → Bool ...
```

- Haskell deals with overloading by a general mechanism called *type classes*. It is considered a major feature of Haskell.
- While the type class is an interesting topic, we might not cover much of it since it is orthogonal to the central message of this course.

4.3 Products

Tuples

- The polymorphic type (a, b) is essentially the same as the following declaration:

```
data Pair a b = MkPair a b
```

- Or, had Haskell allow us to use symbols:

```
data (a, b) = (a, b)
```

- Two projections:

```
fst      :: (a, b) → a
fst (a, b) = a
snd      :: (a, b) → b
snd (a, b) = b
```

5 Functions on Lists

Lists in Haskell

- Traditionally an important datatype in functional languages.
- In Haskell, all elements in a list must be of the same type.
 - $[1, 2, 3, 4] :: [Int]$
 - $[True, False, True] :: [Bool]$
 - $[[1, 2], [], [6, 7]] :: [[Int]]$
 - $[] :: [a]$, the empty list (whose element type is not determined).
- *String* is an abbreviation for $[Char]$; "abcd" is an abbreviation of $['a', 'b', 'c', 'd']$.

List as a Datatype

- $[] :: [a]$ is the empty list whose element type is not determined.
- If a list is non-empty, the leftmost element is called its *head* and the rest its *tail*.
- The constructor $(:) :: a \rightarrow [a] \rightarrow [a]$ builds a list. E.g. in $x : xs$, x is the head and xs the tail of the new list.
- You can think of a list as being defined by


```
data [a] = [] | a : [a]
```

 - $[1, 2, 3]$ is an abbreviation of $1 : (2 : (3 : []))$.

Head and Tail

- $head :: [a] \rightarrow a$. e.g. $head [1, 2, 3] = 1$.
- $tail :: [a] \rightarrow [a]$. e.g. $tail [1, 2, 3] = [2, 3]$.
- $init :: [a] \rightarrow [a]$. e.g. $init [1, 2, 3] = [1, 2]$.
- $last :: [a] \rightarrow a$. e.g. $last [1, 2, 3] = 3$.
- They are all partial functions on non-empty lists. e.g. $head [] = \perp$.
- $null :: [a] \rightarrow Bool$ checks whether a list is empty.

```
null []      = True
null (x : xs) = False
```

5.1 List Generation

List Generation

- $[0..25]$ generates the list $[0, 1, 2..25]$.
- $[0, 2..25]$ yields $[0, 2, 4..24]$.
- $[2..0]$ yields $[\]$.
- The same works for all *ordered* types. For example *Char*:
 - $['a'..'z']$ yields $['a', 'b', 'c'..'z']$.
- $[1..]$ yields the *infinite* list $[1, 2, 3..]$.

List Comprehension

- Some functional languages provide a convenient notation for list generation. It can be defined in terms of simpler functions.
- e.g. $[x \times x \mid x \leftarrow [1..5], \text{odd } x] = [1, 9, 25]$.
- Syntax: $[e \mid Q_1, Q_2..]$. Each Q_i is either
 - a generator $x \leftarrow xs$, where x is a (local) variable or pattern of type a while xs is an expression yielding a list of type $[a]$, or
 - a guard, a boolean valued expression (e.g. *odd* x).
 - e is an expression that can involve new local variables introduced by the generators.

List Comprehension

Examples:

- $[(a, b) \mid a \leftarrow [1..3], b \leftarrow [1..2]] = [(1, 1), (1, 2), (2, 1), (2, 2), (3, 1), (3, 2)]$
- $[(a, b) \mid b \leftarrow [1..2], a \leftarrow [1..3]] = [(1, 1), (2, 1), (3, 1), (1, 2), (2, 2), (3, 2)]$
- $[(i, j) \mid i \leftarrow [1..4], j \leftarrow [i + 1..4]] = [(1, 2), (1, 3), (1, 4), (2, 3), (2, 4), (3, 4)]$
- $[(i, j) \mid i \leftarrow [1..4], \text{even } i, j \leftarrow [i + 1..4], \text{odd } j] = [(2, 3)]$

5.2 Combinators on Lists

Two Modes of Programming

- Functional programmers switch between two modes of programming.
 - Inductive/recursive mode: go into the structure of the input data and recursively process it.
 - Combinatorial mode: compose programs using existing functions (combinators), process the input in stages.
- We will try the latter style today. However, that means we have to familiarise ourselves to a large collection of library functions.
- In the next lecture we will talk about how these library functions can be defined, in the former style.

Length and Indexing

- $(!!) :: [a] \rightarrow Int \rightarrow a$. List indexing starts from zero. e.g. $[1, 2, 3]!!0 = 1$.
- $length :: [a] \rightarrow Int$. e.g. $length [0..9] = 10$.

Append and Concatenation

- Append: $(+) :: [a] \rightarrow [a] \rightarrow [a]$. In ASCII one types $(++)$.
 - $[1, 2] + [3, 4, 5] = [1, 2, 3, 4, 5]$
 - $[\] + [3, 4, 5] = [3, 4, 5] = [3, 4, 5] + [\]$
- Compare with $(:) :: a \rightarrow [a] \rightarrow [a]$. It is a type error to write $[\] : [3, 4, 5]$. $(+)$ is defined in terms of $(:)$.
- $concat :: [[a]] \rightarrow [a]$.
 - e.g. $concat [[1, 2], [\], [3, 4], [5]] = [1, 2, 3, 4, 5]$.
 - $concat$ is defined in terms of $(+)$.

Take and Drop

- $take\ n$ takes the first n elements of the list. For a definition:

$$\begin{aligned} take & :: Int \rightarrow [a] \rightarrow [a] \\ take\ 0\ xs & = [\] \\ take\ (n + 1)\ [\] & = [\] \\ take\ (n + 1)\ (x : xs) & = x : take\ n\ xs \end{aligned}$$

- For example, *take* 0 *xs* = []
- *take* 3 "abcde" = "abc"
- *take* 3 "ab" = "ab"

- Working with infinite list: *take* 5 [1..] = [1, 2, 3, 4, 5]. Thanks to normal order (lazy) evaluation.
- Dually, *drop* *n* drops the first *n* elements of the list. For a definition:

```
drop           :: Int -> [a] -> [a]
drop 0 xs     = xs
drop (n + 1) [] = []
drop (n + 1) (x : xs) = drop n xs
```

- For example, *drop* 0 *xs* = *xs*
- *drop* 3 "abcde" = "cd"
- *drop* 3 "ab" = ""

- *take* *n* *xs* + *drop* *n* *xs* = *xs*, as long as *n* ≠ ⊥.

Map and λ

- *map* :: (a → b) → [a] → [b]. e.g. *map* (1+) [1, 2, 3, 4, 5] = [2, 3, 4, 5, 6].
- *map square* [1, 2, 3, 4] = [1, 4, 9, 16].
- Every once in a while you may need a small function which you do not want to give a name to. At such moments you can use the λ notation:
 - *map* (λx → x × x) [1, 2, 3, 4] = [1, 4, 9, 16]
 - In ASCII λ is written \.
- λ is an important primitive notion. We will talk more about it later.

Filter

- *filter* :: (a → Bool) → [a] → [a].
 - e.g. *filter even* [2, 7, 4, 3] = [2, 4]
 - *filter* (λn → n ‘mod’ 3 == 0) [3, 2, 6, 7] = [3, 6]
- Application: count the number of occurrences of ‘a’ in a list:
 - *length* · *filter* (‘a’ ==)
 - Or *length* · *filter* (λx → ‘a’ == x)
- **Note** a list comprehension can always be translated into a combination of primitive list generators and *map*, *filter*, and *concat*.

Zip

- *zip* :: [a] → [b] → [(a, b)]
- e.g. *zip* "abcde" [1, 2, 3] = [(‘a’, 1), (‘b’, 2), (‘c’, 3)]
- The length of the resulting list is the length of the shorter input list.

Positions

- Exercise: define *positions* :: Char → String → [Int], such that *positions* *x* *xs* returns the positions of occurrences of *x* in *xs*. E.g. *positions* ‘o’ "roodo" = [1, 2, 4].
- *positions* *x* *xs* = *map snd* (*filter* ((*x* ==) · *fst*) (*zip* *xs* [0..]))
- Or, *positions* *x* *xs* = *map snd* (*filter* (λ(y, i) → *x* == *y*) (*zip* *xs* [0..]))
- What if you want only the position of the *first* occurrence of *x*?

```
pos           :: Char -> String -> Int
pos x xs     = head (positions x xs)
```

- Due to lazy evaluation (normal order evaluation), positions of the other occurrences are *not* evaluated!
- **Note** For now, think of “lazy evaluation” as another (more popular) name for normal order evaluation. Some people distinguish them by saying that normal order evaluation is a mathematical idea while lazy evaluation is a way to implement normal order evaluation.

Morals of the Story

- Lazy evaluation helps to improve modularity.
 - List combinators can be conveniently reused. Only the relevant parts are computed.
- The combinator style encourages “wholemeal programming”.
 - Think of aggregate data as a whole, and process them as a whole!

6 λ expressions

- $\lambda x \rightarrow e$ denotes a function whose argument is x and whose body is e .
- $(\lambda x \rightarrow e_1) e_2$ denotes the function $(\lambda x \rightarrow e_1)$ applied to e_2 . It can be reduced to e_1 with its free occurrences of x replaced by e_2 .
- E.g.

$$\begin{aligned} & (\lambda x \rightarrow x \times x) (3 + 4) \\ = & (3 + 4) \times (3 + 4) \\ = & 49 . \end{aligned}$$

- λ expression is a primitive and essential notion. Many other constructs can be seen as syntax sugar of λ 's.
- For example, our previous definition of *square* can be seen as an abbreviation of

$$\begin{aligned} \text{square} & :: \text{Int} \rightarrow \text{Int} \\ \text{square} & = \lambda x \rightarrow x \times x . \end{aligned}$$

- Indeed, *square* is merely a value that happens to be a function, which is in turn given by a λ expression.
- λ 's are like all values — they can appear inside an expression, be passed as parameters, returned as results, etc.
- In fact, it is possible to build a complete programming language consisting of only λ expressions and applications. Look up “ λ calculus”.
- $\lambda x \rightarrow \lambda y \rightarrow e$ is abbreviated to $\lambda x y \rightarrow e$.
- The following definitions are all equivalent:

$$\begin{aligned} \text{smaller } x y & = \mathbf{if } x \leq y \mathbf{ then } x \mathbf{ else } y \\ \text{smaller } x & = \lambda y \rightarrow \mathbf{if } x \leq y \mathbf{ then } x \mathbf{ else } y \\ \text{smaller} & = \lambda x \rightarrow \lambda y \rightarrow \mathbf{if } x \leq y \mathbf{ then } x \mathbf{ else } y \\ \text{smaller} & = \lambda x y \rightarrow \mathbf{if } x \leq y \mathbf{ then } x \mathbf{ else } y . \end{aligned}$$

7 Fold on Lists

Replacing Constructors

- The function *foldr* is among the most important functions on lists.

$$\text{foldr} :: (a \rightarrow b \rightarrow b) \rightarrow b \rightarrow [a] \rightarrow b$$

- One way to look at *foldr* $(\oplus) e$ is that it replaces $[]$ with e and $(:)$ with (\oplus) :

$$\begin{aligned} & \text{foldr } (\oplus) e [1, 2, 3, 4] \\ = & \text{foldr } (\oplus) e (1 : (2 : (3 : (4 : [])))) \\ = & 1 \oplus (2 \oplus (3 \oplus (4 \oplus e))) . \end{aligned}$$

- $\text{sum} = \text{foldr } (+) 0$.
- One can see that $\text{id} = \text{foldr } (:) []$.

Some Trivial Folds on Lists

- Function *maximum* returns the maximum element in a list:

$$\text{maximum} = \text{foldr } \max \text{ } -\infty .$$

- Function *prod* returns the product of a list:

$$\text{prod} = \text{foldr } (\times) 1 .$$

- Function *and* returns the conjunction of a list:

$$\text{and} = \text{foldr } (\&\&) \text{ True} .$$

- Lets emphasise again that *id* on lists is a fold:

$$\text{id} = \text{foldr } (:) [] .$$

Some Slightly Complex Folds

- $\text{length} = \text{foldr } (\lambda x n \rightarrow 1 + n) 0$.
- $\text{map } f = \text{foldr } (\lambda x xs \rightarrow f x : xs) []$.
- $xs \# ys = \text{foldr } (:) ys xs$. Compare this with *id*!
- $\text{filter } p = \text{foldr } (\text{fil } p) []$ where $\text{fil } p x xs = \mathbf{if } p x \mathbf{ then } (x : xs) \mathbf{ else } xs$.

The Ubiquitous Fold

- In fact, *any* function that takes a list as its input can be written in terms of *foldr* — although it might not be always practical.
- With fold it comes one of the most important theorem in program calculation — the fold-fusion theorem. We might not have time to cover it, though.

8 Induction on Natural Numbers

Total Functional Programming

- The next few lectures concerns inductive definitions and proofs of datatypes and programs.
- While Haskell provides allows one to define non-terminating functions, infinite data structures, for now we will only consider its total, finite fragment.
- That is, we temporarily
 - consider only finite data structures,
 - demand that functions terminate for all value in its input type, and
 - provide guidelines to construct such functions.
- Infinite datatypes and non-termination will be discussed later in this course.

The So-Called “Mathematical Induction”

- Let P be a predicate on natural numbers.
 - What is a predicate? Such a predicate can be seen as a function of type $\mathbb{N} \rightarrow Bool$.
 - So far, we see Haskell functions as simple mathematical functions too.
 - However, Haskell functions will turn out to be more complex than mere mathematical functions later. To avoid confusion, we do not use the notation $\mathbb{N} \rightarrow Bool$ for predicates.
- We’ve all learnt this principle of proof by induction: to prove that P holds for all natural numbers, it is sufficient to show that
 - $P\ 0$ holds;
 - $P\ (1 + n)$ holds provided that $P\ n$ does.

Proof by Induction on Natural Numbers

- We can see the above inductive principle as a result of seeing natural numbers as defined by the datatype ¹

data $\mathbb{N} = 0 \mid 1 + \mathbb{N}$.

¹Not a real Haskell definition.

- That is, any natural number is either 0, or $1 + n$ where n is a natural number.
- The type \mathbb{N} is the *smallest* set such that
 1. 0 is in \mathbb{N} ;
 2. if n is in \mathbb{N} , so is $1 + n$.
- Thus to show that P holds for all natural numbers, we only need to consider these two cases.
- In this lecture, $1 +$ is written in bold font to emphasise that it is a data constructor (as opposed to the function $(+)$, to be defined later, applied to a number 1).

Inductively Defined Functions

- Since the type \mathbb{N} is defined by two cases, it is natural to define functions on \mathbb{N} following the structure:

$exp \quad \quad \quad :: \mathbb{N} \rightarrow \mathbb{N} \rightarrow \mathbb{N}$
 $exp\ b\ 0 \quad \quad = 1$
 $exp\ b\ (1 + n) = b \times exp\ b\ n$.

- Even addition can be defined inductively

$(+)$ $\quad \quad \quad :: \mathbb{N} \rightarrow \mathbb{N} \rightarrow \mathbb{N}$
 $0 + n \quad \quad \quad = n$
 $(1 + m) + n = 1 + (m + n)$.

- Exercise: define (\times) ?

Without the $n + k$ Pattern

- Unfortunately, newer versions of Haskell abandoned the “ $n + k$ pattern” used in the previous slide. And there is not a built-in type for \mathbb{N} . Instead we have to write:

$exp \quad \quad \quad :: Int \rightarrow Int \rightarrow Int$
 $exp\ b\ 0 = 1$
 $exp\ b\ n = b \times exp\ b\ (n - 1)$.

- For the purpose of this course, the pattern $1 + n$ reveals the correspondence between \mathbb{N} and lists, and matches our proof style. Thus we will use it in the lecture.
- Remember to remove them in your code.

Proof by Induction

- To prove properties about \mathbb{N} , we follow the structure as well.
- E.g. to prove that $\text{exp } b (m + n) = \text{exp } b m \times \text{exp } b n$.
- One possibility is to perform induction on m . That is, prove Pm for all $m :: \mathbb{N}$, where $Pm \equiv \text{exp } b (m + n) = \text{exp } b m \times \text{exp } b n$.

Case $m := 0$:

$$\begin{aligned} & \text{exp } b (0 + n) \\ = & \quad \{ \text{defn. of } (+) \} \\ & \text{exp } b n \\ = & \quad \{ \text{defn. of } (\times) \} \\ & 1 \times \text{exp } b n \\ = & \quad \{ \text{defn. of } \text{exp} \} \\ & \text{exp } b 0 \times \text{exp } b n . \end{aligned}$$

Proof by Induction

Case $m := 1 + m$:

$$\begin{aligned} & \text{exp } b ((1 + m) + n) \\ = & \quad \{ \text{defn. of } (+) \} \\ & \text{exp } b (1 + (m + n)) \\ = & \quad \{ \text{defn. of } \text{exp} \} \\ & b \times \text{exp } b (m + n) \\ = & \quad \{ \text{induction} \} \\ & b \times (\text{exp } b m \times \text{exp } b n) \\ = & \quad \{ (\times) \text{ associative} \} \\ & (b \times \text{exp } b m) \times \text{exp } b n \\ = & \quad \{ \text{defn. of } \text{exp} \} \\ & \text{exp } b (1 + m) \times \text{exp } b n . \end{aligned}$$

Structure Proofs by Programs

- The inductive proof could be carried out smoothly, because both $(+)$ and exp are defined inductively on its lefthand argument (of type \mathbb{N}).
- The structure of the proof follows the structure of the program, which in turns follows the structure of the datatype the program is defined on.

Lists and Natural Numbers

- We have yet to prove that (\times) is associative.

- The proof is quite similar to the proof for associativity of $(+)$, which we will talk about later.
- In fact, \mathbb{N} and lists are closely related in structure.
- Most of us are used to think of numbers as atomic and lists as structured data. Neither is necessarily true.
- For the rest of the course we will demonstrate induction using lists, while taking the properties for \mathbb{N} as given.

9 Induction on Lists

Inductively Defined Lists

- Recall that a (finite) list can be seen as a datatype defined by:²

data $[a] = [] \mid a : [a]$.

- Every list is built from the base case $[]$, with elements added by $(:)$ one by one: $[1, 2, 3] = 1 : (2 : (3 : []))$.
- The type $[a]$ is the *smallest* set such that
 1. $[]$ is in $[a]$;
 2. if xs is in $[a]$ and x is in a , $x : xs$ is in $[a]$ as well.
- But what about infinite lists?
 - For now let's consider finite lists only, as having infinite lists make the *semantics* much more complicated.³
 - In fact, all functions we talk about today are total functions. No \perp involved.

Inductively Defined Functions on Lists

- Many functions on lists can be defined according to how a list is defined:

$sum :: [Int] \rightarrow Int$
 $sum [] = 0$
 $sum (x : xs) = x + sum xs$.

$map :: (a \rightarrow b) \rightarrow [a] \rightarrow [b]$
 $map f [] = []$
 $map f (x : xs) = f x : map f xs$.

²Not a real Haskell definition.

³What does that mean? We will talk about it later.

- $sum [1..10] = 55$
- $map (1+) [1, 2, 3, 4] = [2, 3, 4, 5]$

9.1 Append, and Some of Its Properties

List Append

- The function $(+)$ appends two lists into one

$$\begin{aligned} (+) &:: [a] \rightarrow [a] \rightarrow [a] \\ [] + ys &= ys \\ (x : xs) + ys &= x : (xs + ys) \end{aligned}$$

- Compare the definition with that of $(+)$!

Proof by Structural Induction on Lists

- Recall that every finite list is built from the base case $[]$, with elements added by $(:)$ one by one.
- The type $[a]$ is the smallest set such that
 1. $[]$ is in $[a]$;
 2. if xs is in $[a]$ and x is in a , $x : xs$ is in $[a]$ as well.
- To prove that some property P holds for all finite lists, we show that
 1. $P []$ holds;
 2. $P (x : xs)$ holds, provided that $P xs$ holds.

Appending is Associative

To prove that $xs + (ys + zs) = (xs + ys) + zs$. Case $xs := []$:

$$\begin{aligned} &[] + (ys + zs) \\ = &\{ \text{defn. of } (+) \} \\ &ys + zs \\ = &\{ \text{defn. of } (+) \} \\ &([] + ys) + zs \end{aligned}$$

Appending is Associative

Case $xs := x : xs$:

$$\begin{aligned} &(x : xs) + (ys + zs) \\ = &\{ \text{defn. of } (+) \} \\ &x : (xs + (ys + zs)) \\ = &\{ \text{induction} \} \\ &x : ((xs + ys) + zs) \\ = &\{ \text{defn. of } (+) \} \\ &(x : (xs + ys)) + zs \\ = &\{ \text{defn. of } (+) \} \\ &((x : xs) + ys) + zs \end{aligned}$$

Length

- The function $length$ defined inductively:

$$\begin{aligned} length &:: [a] \rightarrow Int \\ length [] &= 0 \\ length (x : xs) &= 1 + length xs \end{aligned}$$

- Exercise: prove that $length$ distributes into $(+)$:

$$length (xs + ys) = length xs + length ys$$

Concatenation

- While $(+)$ repeatedly applies $(:)$, the function $concat$ repeatedly calls $(+)$:

$$\begin{aligned} concat &:: [[a]] \rightarrow [a] \\ concat [] &= [] \\ concat (xs : xss) &= xs + concat xss \end{aligned}$$

- Compare with sum .
- Exercise: prove $sum \cdot concat = sum \cdot map \cdot sum$.

9.2 More Inductively Defined Functions

Definition by Induction/Recursion

- Rather than giving commands, in functional programming we specify values; instead of performing repeated actions, we define values on inductively defined structures.
- Thus induction (or in general, recursion) is the only “control structure” we have. (We do identify and abstract over plenty of patterns of recursion, though.)
- **Note** Terminology: an inductive definition, as we have seen, define “bigger” things in terms of “smaller” things. Recursion, on the other hand, is a more general term, meaning “to define one entity in terms of itself.”
- To inductively define a function f on lists, we specify a value for the base case ($f []$) and, assuming that $f xs$ has been computed, consider how to construct $f (x : xs)$ out of $f xs$.

Filter

- *filter* p xs keeps only those elements in xs that satisfy p .

$$\begin{aligned} \text{filter} &:: (a \rightarrow \text{Bool}) \rightarrow [a] \rightarrow [a] \\ \text{filter } p \ [] &= [] \\ \text{filter } p (x : xs) &| p \ x = x : \text{filter } p \ xs \\ &| \text{otherwise} = \text{filter } p \ xs \ . \end{aligned}$$

Take and Drop

- Recall *take* and *drop*, which we used in the previous exercise.

$$\begin{aligned} \text{take} &:: \text{Int} \rightarrow [a] \rightarrow [a] \\ \text{take } 0 \ xs &= [] \\ \text{take } (1 + n) \ [] &= [] \\ \text{take } (1 + n) (x : xs) &= x : \text{take } n \ xs \ . \end{aligned}$$
$$\begin{aligned} \text{drop} &:: \text{Int} \rightarrow [a] \rightarrow [a] \\ \text{drop } 0 \ xs &= xs \\ \text{drop } (1 + n) \ [] &= [] \\ \text{drop } (1 + n) (x : xs) &= \text{drop } n \ xs \ . \end{aligned}$$

- Prove: $\text{take } n \ xs \ ++ \ \text{drop } n \ xs = xs$, for all n and xs .

TakeWhile and DropWhile

- *takeWhile* p xs yields the longest prefix of xs such that p holds for each element.

$$\begin{aligned} \text{takeWhile} &:: (a \rightarrow \text{Bool}) \rightarrow [a] \rightarrow [a] \\ \text{takeWhile } p \ [] &= [] \\ \text{takeWhile } p (x : xs) &| p \ x = x : \text{takeWhile } p \ xs \\ &| \text{otherwise} = [] \ . \end{aligned}$$

- *dropWhile* p xs drops the prefix from xs .

$$\begin{aligned} \text{dropWhile} &:: (a \rightarrow \text{Bool}) \rightarrow [a] \rightarrow [a] \\ \text{dropWhile } p \ [] &= [] \\ \text{dropWhile } p (x : xs) &| p \ x = \text{dropWhile } p \ xs \\ &| \text{otherwise} = x : xs \ . \end{aligned}$$

- Prove: $\text{takeWhile } p \ xs \ ++ \ \text{dropWhile } p \ xs = xs$.

List Reversal

- $\text{reverse } [1, 2, 3, 4] = [4, 3, 2, 1]$.

$$\begin{aligned} \text{reverse} &:: [a] \rightarrow [a] \\ \text{reverse} \ [] &= [] \\ \text{reverse} (x : xs) &= \text{reverse } xs \ ++ [x] \ . \end{aligned}$$

All Prefixes and Suffixes

- $\text{inits } [1, 2, 3] = [[], [1], [1, 2], [1, 2, 3]]$

$$\begin{aligned} \text{inits} &:: [a] \rightarrow [[a]] \\ \text{inits} \ [] &= [[]] \\ \text{inits} (x : xs) &= [] : \text{map } (x :) (\text{inits } xs) \ . \end{aligned}$$

- $\text{tails } [1, 2, 3] = [[1, 2, 3], [2, 3], [3], []]$

$$\begin{aligned} \text{tails} &:: [a] \rightarrow [[a]] \\ \text{tails} \ [] &= [[]] \\ \text{tails} (x : xs) &= (x : xs) : \text{tails } xs \ . \end{aligned}$$

Totality

- Structure of our definitions so far:

$$\begin{aligned} f \ [] &= \dots \\ f (x : xs) &= \dots f \ xs \dots \end{aligned}$$

- Both the empty and the non-empty cases are covered, guaranteeing there is a matching clause for all inputs.
- The recursive call is made on a “smaller” argument, guaranteeing termination.

- Together they guarantee that every input is mapped to some output. Thus they define *total* functions on lists.

9.3 Other Patterns of Induction

Variations with the Base Case

- Some functions discriminate between several base cases. E.g.

$$\begin{aligned} \text{fib} &:: \mathbb{N} \rightarrow \mathbb{N} \\ \text{fib } 0 &= 0 \\ \text{fib } 1 &= 1 \\ \text{fib } (2 + n) &= \text{fib } (1 + n) + \text{fib } n \ . \end{aligned}$$

- Some functions make more sense when it is defined only on non-empty lists:

$$\begin{aligned} f [x] &= \dots \\ f (x : xs) &= \dots \end{aligned}$$

- What about totality?

- They are in fact functions defined on a different datatype:

$$\text{data } [a]^+ = \text{Singleton } a \mid a : [a]^+ \ .$$

- We do not want to define *map*, *filter* again for $[a]^+$. Thus we reuse $[a]$ and pretend that we were talking about $[a]^+$.
- It's the same with \mathbb{N} . We embedded \mathbb{N} into *Int*.
- Ideally we'd like to have some form of *subtyping*. But that makes the type system more complex.

Lexicographic Induction

- It also occurs often that we perform *lexicographic induction* on multiple arguments: some arguments decrease in size, while others stay the same.
- E.g. the function *merge* merges two sorted lists into one sorted list:

```
merge :: [Int] -> [Int] -> [Int]
merge [] [] = []
merge [] (y : ys) = y : ys
merge (x : xs) [] = x : xs
merge (x : xs) (y : ys) | x <= y = x : merge xs (y : ys)
                        | otherwise = y : merge (x : xs) ys .
```

Zip

Another example:

```
zip :: [a] -> [b] -> [(a,b)]
zip [] [] = []
zip [] (y : ys) = []
zip (x : xs) [] = []
zip (x : xs) (y : ys) = (x,y) : zip xs ys .
```

Non-Structural Induction

- In most of the programs we've seen so far, the recursive call are made on direct sub-components of the input (e.g. $f(x : xs) = ..f xs..$). This is called *structural induction*.
 - It is relatively easy for compilers to recognise structural induction and determine that a program terminates.
- In fact, we can be sure that a program terminates if the arguments get “smaller” under some (well-founded) ordering.

Mergesort

- In the implementation of mergesort below, for example, the arguments always get smaller in size.

```
msortBy :: [Int] -> [Int]
msortBy [] = []
msortBy [x] = [x]
msortBy xs = merge (msortBy ys) (msortBy zs) ,
  where n = length xs `div` 2
        ys = take n xs
        zs = drop n xs .
```

- What if we omit the case for $[x]$?

- If all cases are covered, and all recursive calls are applied to smaller arguments, the program defines a total function.

A Non-Terminating Definition

- Example of a function, where the argument to the recursive does not reduce in size:

```
f :: Int -> Int
f 0 = 0
f n = f n .
```

- Certainly *f* is not a total function. Do such definitions “mean” something? We will talk about these later.

10 User Defined Inductive Datatypes

Internally Labelled Binary Trees

- This is a possible definition of internally labelled binary trees:

```
data Tree a = Null | Node a (Tree a) (Tree a) ,
```

- on which we may inductively define functions:

```
sumT :: Tree N -> N
sumT Null = 0
sumT (Node x t u) = x + sumT t + sumT u .
```

Exercise: given $(\downarrow) :: \mathbb{N} \rightarrow \mathbb{N} \rightarrow \mathbb{N}$, which yields the smaller one of its arguments, define the following functions

1. $minT :: Tree \mathbb{N} \rightarrow \mathbb{N}$, which computes the minimal element in a tree.

2. $mapT :: (a \rightarrow b) \rightarrow Tree\ a \rightarrow Tree\ b$, which applies the functional argument to each element in a tree.
3. Can you define (\downarrow) inductively on \mathbb{N} ?⁴

Induction Principle for *Tree*

- What is the induction principle for *Tree*?
- To prove that a predicate P on *Tree* holds for every tree, it is sufficient to show that
 1. $P\ \text{Null}$ holds, and;
 2. for every x , t , and u , if $P\ t$ and $P\ u$ holds, $P\ (\text{Node}\ x\ t\ u)$ holds.
- Exercise: prove that for all n and t , $minT\ (mapT\ (n+)\ t) = n + minT\ t$. That is, $minT \cdot mapT\ (n+) = (n+) \cdot minT$.

Induction Principle for Other Types

- Recall that **data** *Bool* = *False* | *True*. Do we have an induction principle for *Bool*?
- To prove a predicate P on *Bool* holds for all booleans, it is sufficient to show that
 1. $P\ \text{False}$ holds, and
 2. $P\ \text{True}$ holds.
- Well, of course.
- What about $(A \times B)$? How to prove that a predicate P on $(A \times B)$ is always true?
- One may prove some property P_1 on A and some property P_2 on B , which together imply P .
- That does not say much. But the “induction principle” for products allows us to extract, from a proof of P , the proofs P_1 and P_2 .
- *Every inductively defined datatype comes with its induction principle.*
- We will come back to this point later.

⁴In the standard Haskell library, (\downarrow) is called *min*.

References

- [Bir98] Richard Simpson Bird. *Introduction to Functional Programming using Haskell*. Prentice Hall, 1998.
- [DI02] Hal Daume III. Yet another haskell tutorial. <http://en.wikibooks.org/wiki/Haskell/YAHT>, 2002.
- [HPF00] Paul Hudak, John Peterson, and Joseph Fasel. A gentle introduction to haskell, version 98. <http://www.haskell.org/tutorial/>, 2000.
- [Hut07] Graham Hutton. *Programming in Haskell*. Cambridge University Press, 2007.
- [Lip11] Miran Lipovača. *Learn You a Haskell for Great Good!* No Starch Press, 2011. Available online at <http://learnyouahaskell.com/>.
- [OSG98] Bryan O’Sullivan, Don Stewart, and John Goerzen. *Real World Haskell*. O’Reilly, 1998. Available online at <http://book.realworldhaskell.org/>.

A GHCi Commands

<code><statement></code>	evaluate/run <code><statement></code>
<code>:</code>	repeat last command
<code>:\{\\n ..lines.. \\n:\\}\\n}</code>	multiline command
<code>:add [*]<module> ...</code>	add module(s) to the current target set
<code>:browse[!] [[*]<mod>]</code>	display the names defined by module <code><mod></code> (!: more details; *: all top-level names)
<code>:cd <dir></code>	change directory to <code><dir></code>
<code>:cmd <expr></code>	run the commands returned by <code><expr>::IO String</code>
<code>:ctags[!] [<file>]</code>	create tags file for Vi (default: "tags") (!: use regex instead of line number)
<code>:def <cmd> <expr></code>	define command <code>:<cmd></code> (later defined command has precedence, <code>::<cmd></code> is always a builtin command)
<code>:edit <file></code>	edit file
<code>:edit</code>	edit last module
<code>:etags [<file>]</code>	create tags file for Emacs (default: "TAGS")
<code>:help, :?</code>	display this list of commands
<code>:info [<name> ...]</code>	display information about the given names
<code>:issafe [<mod>]</code>	display safe haskell information of module <code><mod></code>
<code>:kind <type></code>	show the kind of <code><type></code>
<code>:load [*]<module> ...</code>	load module(s) and their dependents
<code>:main [<arguments> ...]</code>	run the main function with the given arguments
<code>:module [+/-] [*]<mod> ...</code>	set the context for expression evaluation
<code>:quit</code>	exit GHCi
<code>:reload</code>	reload the current module set
<code>:run function [<arguments> ...]</code>	run the function with the given arguments
<code>:script <filename></code>	run the script <code><filename></code>
<code>:type <expr></code>	show the type of <code><expr></code>
<code>:undef <cmd></code>	undefine user-defined command <code>:<cmd></code>
<code>!:<command></code>	run the shell command <code><command></code>

Commands for debugging

<code>:abandon</code>	at a breakpoint, abandon current computation
<code>:back</code>	go back in the history (after <code>:trace</code>)
<code>:break [<mod>] <l> [<col>]</code>	set a breakpoint at the specified location
<code>:break <name></code>	set a breakpoint on the specified function
<code>:continue</code>	resume after a breakpoint
<code>:delete <number></code>	delete the specified breakpoint
<code>:delete *</code>	delete all breakpoints
<code>:force <expr></code>	print <code><expr></code> , forcing unevaluated parts
<code>:forward</code>	go forward in the history (after <code>:back</code>)
<code>:history [<n>]</code>	after <code>:trace</code> , show the execution history
<code>:list</code>	show the source code around current breakpoint
<code>:list identifier</code>	show the source code for <code><identifier></code>
<code>:list [<module>] <line></code>	show the source code around line number <code><line></code>
<code>:print [<name> ...]</code>	prints a value without forcing its computation
<code>:sprint [<name> ...]</code>	simplified version of <code>:print</code>
<code>:step</code>	single-step after stopping at a breakpoint
<code>:step <expr></code>	single-step into <code><expr></code>

<code>:steplocal</code>	single-step within the current top-level binding
<code>:stepmodule</code>	single-step restricted to the current module
<code>:trace</code>	trace after stopping at a breakpoint
<code>:trace <expr></code>	evaluate <expr> with tracing on (see <code>:history</code>)

Commands for changing settings

<code>:set <option> ...</code>	set options
<code>:seti <option> ...</code>	set options for interactive evaluation only
<code>:set args <arg> ...</code>	set the arguments returned by <code>System.getArgs</code>
<code>:set prog <progname></code>	set the value returned by <code>System.getProgName</code>
<code>:set prompt <prompt></code>	set the prompt used in GHCi
<code>:set editor <cmd></code>	set the command used for <code>:edit</code>
<code>:set stop [<n>] <cmd></code>	set the command to run when a breakpoint is hit
<code>:unset <option> ...</code>	unset options

Options for `:set` and `:unset`

<code>+m</code>	allow multiline commands
<code>+r</code>	revert top-level expressions after each evaluation
<code>+s</code>	print timing/memory stats after each evaluation
<code>+t</code>	print type after evaluation
<code>-<flags></code>	most GHC command line flags can also be set here (eg. <code>-v2</code> , <code>-fglasgow-exts</code> , etc). For GHCi-specific flags, see User's Guide, Flag reference, Interactive-mode options.

Commands for displaying information

<code>:show bindings</code>	show the current bindings made at the prompt
<code>:show breaks</code>	show the active breakpoints
<code>:show context</code>	show the breakpoint context
<code>:show imports</code>	show the current imports
<code>:show modules</code>	show the currently loaded modules
<code>:show packages</code>	show the currently active package flags
<code>:show language</code>	show the currently active language flags
<code>:show <setting></code>	show value of <setting>, which is one of [args, prog, prompt, editor, stop]
<code>:showi language</code>	show language flags for interactive evaluation